

Relaxation-encoded NMR experiments for mixture analysis:

REST and beer

Electronic Supporting Information

*Guilherme Dal Poggetto, Laura Castañar, Ralph W. Adams, Gareth A. Morris and
Mathias Nilsson**

Table of contents

A. Experimental section

1. Pulse sequence details
2. Data processing using the DOSY Toolbox
 - 2.1. Relaxation module (ROSY)
 - 2.2. Multivariate methods (SCORE/OUTSCORE)

B. Experimental data

1. Lager beer sample in D₂O.
2. Mixture of lactose and melibiose in D₂O.

C. Pulse sequences (Bruker)

1. Inversion recovery (IR) with CW presaturation (for T_1 measurement)
2. Periodic refocussing of J evolution by coherence transfer (PROJECT) with CW presaturation (for T_2 measurement)
3. IR REST₁ with CW presaturation
4. PROJECT REST₂ with CW presaturation
5. IR REST₁ (no presaturation)
6. PROJECT REST₂ (no presaturation)

A. Experimental section

All spectra were recorded on a Bruker Avance II+ 500 MHz spectrometer with a 5 mm BBO probe equipped with a z-gradient coil with a maximum nominal gradient strength of 53 G cm⁻¹.

1. Pulse sequence

The detailed pulse sequences for the relaxation-encoded selective-TOCSY (REST) experiments are shown in Fig. ESI1. Black narrow and grey wide filled rectangles represent hard 90° and 180° pulses, respectively. The selective 180° pulse, represented by the shaped wide grey pulse, is applied to the signal area of interest, and typically RSNOB or REBURP shapes are used. In order to be able to use reference deconvolution with the TSP-*d*₄ singlet, the selective pulse needs to be shaped to refocus both the resonance of interest and TSP. The TOCSY transfer is achieved by using the DIPSI-2 mixing scheme with a mixing time of 50-200 ms depending on the spin system. Arrowed trapezoids on either side of the DIPSI-2 isotropic mixing element are low-power 180° chirp pulses of 20 kHz bandwidth, used to suppress zero quantum coherences, and their durations were set to 10 and 30 ms. Gradient pulse G₁ (amplitude of 12.4 G cm⁻¹, duration of 1 ms followed by a recovery delay of at least 200 μs) is used to eliminate magnetization of signals outside of the bandwidth of the selective pulse. G₂ and G₄ (amplitudes of 1.6 and 1.1 G cm⁻¹, respectively) are weak pulsed field gradients applied simultaneously with the chirp pulses to suppress zero quantum coherences. G₃ is a spoil gradient pulse with an amplitude of 16.6 G cm⁻¹ and duration of 1-2 ms. In IR REST₁, τ₁ is an incremented delay, defined in the VD list (user-defined), and should contain at least 8 delays for good exponential fitting. The largest value of τ₁ should be long enough for the magnetization to relax fully. In PROJECT REST₂, τ₂ should have a duration at least 50 times that of the hard 180° pulse, to limit sample heating, but still be small enough to avoid coupling evolution (i.e. τ₂ << 1/J). For all the experiments demonstrated here, τ₂ was set to 1.1 ms. The total echo time is defined by the total duration of each perfect echo element (4 τ₂ + 5 p1) multiplied by the loop counter (n) defined in the VC list, which should contain at least 8 values for good exponential fitting. The minimum phase cycle recommended is 16 steps for REST₁ and 8 for REST₂; the full phase cycle is given in Table S1. Low power continuous wave irradiation can be used for water presaturation during d1, which provides good results for PROJECT REST₂ but less good for IR REST₁. In the latter case, continuous wave irradiation can also be applied during the incrementing delay τ₁ for more efficient water suppression.




Fig. ESI 1 – Pulse sequences. a) Inversion recovery, b) PROJECT and c) REST, using e.g. the sequence elements of (a) or (b) for relaxation encoding.

Table S1 – Phase cycling of REST experiments

	REST ₁	REST ₂
Φ ₀	(x ₄ , -x ₄) ₄	-
Φ ₁	(x ₈ , -x ₈) ₂	(x ₄ , -x ₄) ₄
Φ ₂	(x, y, -x, -y) ₈	(x, y, -x, -y) ₈
Φ ₃	x ₃₂	x ₃₂
Φ ₄	(x ₁₆ , y ₁₆)	(x ₈ , y ₈ , -x ₈ , -y ₈)
Φ _R	(x, -x) ₄ , (-x, x) ₄ , (y, -y) ₄ , (-y, y) ₄	(x, -x) ₂ , (-x, x) ₂ , (y, -y) ₂ , (-y, y) ₂ (-x, x) ₂ , (x, -x) ₂ , (-y, y) ₂ , (y, -y) ₂

2. Data processing using the DOSY Toolbox

The latest version of the DOSY Toolbox, which includes relaxation processing, can be downloaded at <http://nmr.chemistry.manchester.ac.uk/?q=node/8>. For more information, or error reporting, please email us. All processed data used in this communication used the Matlab version of the toolbox, which is compatible with Matlab 2015b (or more recent). The computational time required for processing will depend on the computer being used.

To process data from any spectrometer manufacturer (Bruker, Varian/Agilent or JEOL) import the data by clicking on “DOSY Files” and “Import”. Either raw or processed data can be imported. Ensure that values from the VD/VC list have been imported correctly by going to “Edit DOSY” and “Relaxation”. The values of any list can be edited in this window. After the 1D spectra have being processed (or

imported as processed data), including any Fourier transformation, reference deconvolution, baseline correction or aligning of spectra, it is possible to display relaxation data (both T_1 and T_2) in two ways: in a pseudo 2D spectrum (ROSY), or using multivariate methods (SCORE/OUTSCORE).

2.1. Relaxation module (ROSY)

In “Advanced Processing” choose “T1/T2”. Zoom in on the desired region and select the threshold for the peak picking routine using “Peak Pick”. For inversion recovery or saturation recovery data set a threshold in the last spectrum (longest τ_1), while in other experiments set a threshold in the first spectrum (smallest VC counter). This step can be skipped if all points of the spectrum are to be used in the relaxation fitting (using “All Frq”). Signal integral, instead of intensity, may also be used to perform the signal fitting (using “Integrals”). The Fit Type (T1 or T2) should be chosen for the particular experiment imported. Clicking on “Run” generates the pseudo 2D spectrum in a new window. Individual experimental spectra can be excluded from the fitting routine if “Prune” is used.

2.2. Multivariate methods (SCORE/OUTSCORE)

In “Advanced Processing” choose “RSCORE”. Zoom in on the desired region. Multivariate methods do not need a threshold setting and use the whole displayed spectrum. The fitting function (T1 or T2) must be chosen for the particular experiment imported. The guess for the relaxation time can be defined as a random value or the value obtained in the fitting. In the second case the VD/VC list must be correct. The user must also select the number of components to be fitted (“N. free”), which for methods such as SCORE and OUTSCORE is in practice limited to 2-5 components, depending on the range of relaxation time constant and on the signal overlap between components. SCORE minimizes the residual errors from the components and OUTSCORE minimizes the cross-talk between them. A non-negativity constraint can be used, which will only consider positive values for relaxation times; this normally increases the number of iterations required. Clicking on “Run” starts iteration. When it finishes, the component 1D spectra will be generated in a new window. Experimental spectra can be excluded from the fitting routine if “Prune” is used.

B. Experimental data

All experimental data for this paper are freely available for download from DOI: 10.15127/1.307570.

1. Lager beer sample in D₂O.

The sample was prepared by mixing 480 μ L of the degassed lager beer with 120 μ L of D₂O with TSP-*d*₄ (approx. 10 mM final concentration). Experiments were run at 298 K. For all experiments except the 2D TOCSY, the ¹H spectral window was set to 7002.80 Hz (14.00 ppm), the carrier frequency to 1500.39 Hz (3.0 ppm) and the duration of the hard 90° pulse was 10.1 μ s. For the 2D TOCSY both dimensions were set to 3500.92 Hz (7.00 ppm) and the carrier frequency to 1250.33 Hz (2.50 ppm).




Fig. ESI 2 – Expanded region of a 500 MHz 2D TOCSY spectrum of lager beer in D₂O. Data were recorded with 8 scans, 2k complex points in the direct dimension and 512 t_1 increments, with experiment time of 4 h. Direct and indirect dimensions were zero-filled to 8k and 1k points, respectively. Lorentz-to-Gauss transformation was applied with LB of -0.01 Hz, GB of 0.001 for the direct dimension and LB of -0.01 Hz, GB of 0.0003 for the indirect dimension.




Fig. ESI 3 – Aliased HSQC spectrum of the lager beer sample in D₂O. Data were recorded with 8 scans, 2k complex points and 256 t_1 increments. Direct and indirect dimensions were zero-filled to 4k and 2k points, respectively, and 90° phase-shifted squared sine-bell apodization was applied in both dimensions prior to Fourier transformation. The ¹³C spectral window was fixed at 377.3 Hz (3.00 ppm) centred on 11569.72 Hz (92.00 ppm). Even with the high indirect dimension resolution here (about 2.9 Hz), the carbon signals remain unresolved, so no new information about sample composition is gained.




Fig. ESI 4 – a) 1D selective TOCSY spectrum for lager beer sample, with water presaturation, selective pulse at 5.229 ppm (RSNOB of duration 50 ms, bandwidth approx. 47 Hz) and mixing time of 100 ms, and b) ¹H 1D spectrum for beer sample, with water presaturation. The selective TOCSY spectrum was recorded with 32k complex points using 32 transients in an experiment time around 4 min 30 s. Prior to Fourier transformation, zero-filling to 128k was applied.




Fig. ESI 5 – a) Inversion recovery ROSY and b) IR REST₁ ROSY (selective pulse at 5.229 ppm) for the lager beer sample (with water presaturation using continuous-wave irradiation). Both spectra were recorded with 32k complex points, using 16 transients for (a) and 32 for (b). Prior to Fourier transformation, zero-filling to 128k was applied. Reference deconvolution with Lorentz to Gauss transformation was applied to both spectra, with a target Lorentzian linewidth of -0.4 Hz and a Gaussian linewidth of 2 Hz for the TSP-*d*₄ signal. The 10 τ_1 increments in the VD list used were: 0.001, 0.04, 0.16, 0.64, 0.96, 1.28, 2.56, 3.84, 5.12 and 7.68 s.

2. Mixture of lactose and melibiose in D₂O.

The sample was prepared by dissolving 20 mg of melibiose and 20 mg of lactose in 600 μ L of D₂O (giving approx. 100 mM for each). Experiments were run at 298 K. For all experiments the ¹H spectral window was set to 7002.80 Hz (14.00 ppm), the carrier frequency to 1500.39 Hz (3.0 ppm) and the duration of the hard 90° pulse was 10.6 μ s.




Fig. ESI 6 – a) 1D selective TOCSY spectrum for disaccharide sample, with water presaturation, selective pulse at 5.239 ppm (RSNOB of duration 30 ms, bandwidth approx. 77 Hz) and mixing time of 120 ms, and b) ¹H 1D spectrum for beer sample, with water presaturation. The selective TOCSY spectrum was recorded with 32k complex points using 32 transients in an experiment time of 8 min 43 s. Prior to Fourier transformation, zero-filling to 128k was applied.




Fig. ESI 7 – a) Inversion recovery-ROSY and b) IR-REST₁-ROSY (selective pulse at 5.239 ppm) for the disaccharide sample (with water presaturation using continuous-wave irradiation). Both spectra were recorded with 32k complex points, using 16 transients for (a) and 32 for (b). Prior to Fourier transformation, zero-filling to 128k was applied. Reference deconvolution with Lorentz to Gauss transformation was applied to both spectra, with a target Lorentzian linewidth of -0.4 Hz and a Gaussian linewidth of 2 Hz for the TSP-*d*₄ signal. The 12 τ_1 increments in the VD list used were: 0.0016, 0.016, 0.08, 0.16, 0.32, 0.64, 0.8, 1.28, 1.92, 2.56, 5.12 and 10.24 s.




Fig. ESI 8 – REST₂ SCORE components for the disaccharide sample in D₂O, fitting for 2 components using non-negativity constraint. a) SCORE component for lactose α -glucose, b) SCORE component for melibiose α -glucose, c) α -glucose reference spectrum using 1D selective TOCSY on a sample of lactose and d) α -glucose reference spectrum using 1D selective TOCSY on a sample of melibiose.

C. Pulse sequences (Bruker format)

1. Inversion recovery (IR) with CW presaturation (for T_1 measurement)

```
;2dt1irpr_cwvd

; T1 measurement using inversion recovery
; with saturation pulses prior to d1 and
; with continuous wave excitation for water presaturation during d1 and VD
;
; Modified 30/01/17
;
;      Guilherme Dal Poggetto and Laura Castañar
;      Univeristy of Manchester
;
;Avance II+/III Version
;Topspin 3.x
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"

"d11=30m"
"d12=20u"

"acqt0=-p1*2/3.1416"
"cnst30=0"

1 ze
2 d11 p1:f1
  10u UNBLKGRAD                ;saturation pulses
  p8 ph21
  p8 ph22
  p18:gp8*-1
  d18 BLKGRAD
  50u LOCKH_OFF
  1m fq=cnst29(bf ppm):f1      ;solvent presaturation
  d12 p19:f1
  d1 cw:f1 ph29
  4u do:f1
  1m fq=cnst30:f1
  d12 p11:f1
  50u LOCKH_ON
  50u UNBLKGRAMP

3 p2 ph1                      ;inversion recovery with solvent saturation during vd
  4u fq=cnst29(bf ppm):f1
  4u p19:f1
  vd cw:f1 ph29
```

```
4u do:f1
4u fq=cnst30:f1
4u pl1:f1
p1 ph2
3u BLKGRAMP
```

```
go=2 ph31
d11 wr #0 if #0 ivd
50u LOCKH_OFF
lo to 1 times td1
exit
```

```
ph1=0 2
ph2=0 0 2 2 1 1 3 3
ph21=0
ph22=1
ph29=0
ph31=0 0 2 2 1 1 3 3
```

```
;POWER LEVEL
```

```
;pl0 : f1 channel - zero power (0W)
;p11 : f1 channel - power level for pulse (default)
;p19 : f1 channel - power level for continuous wave pulse
```

```
;PULSE DURATION
```

```
;p1 : f1 channel - 90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p8 : f1 channel - saturation pulse [1-1.5 msec]
```

```
;GRADIENT DURATION
```

```
;p18 : saturation gradient pulse [1 msec]
```

```
;DELAY
```

```
;d1 : relaxation delay; 1-5 * T1 [2-10 s]
;d11 : delay for disk I/O [30 msec]
;d18 : recovery delay for saturation gradient [1 ms]
;vd : variable delay, taken from vd-list
```

```
;GRADIENT SHAPE
```

```
;gpnam8 : SMSQ10.100
```

```
;GRADIENT STRENGTH
```

```
;gpz8 : saturation recovery gradient [13%]
```

```
;CONSTANTS
```

```
;cnst29: water chemical shift (in ppm)
```

```
;OTHER
```

```
;NS: 8 * n, total number of scans
```

```
;DS: 4
```

```
;td1 : number of delays in VDLIST (number of experiments)
```

```
;FnMODE: QF
```

```
;define VDLIST
```

```
;this pulse program produces a ser-file (PARMOD = 2D)
```

2. Periodic refocussing of J evolution by coherence transfer (PROJECT) with CW presaturation (for T_2 measurement)

;2dt2projpr

; T2 measurement using PROJECT and
; with saturation pulses prior to d1 and
; with continuous wave excitation for water presaturation during d1

;
; Modified 30/01/17

;
; Guilherme Dal Poggetto and Laura Castañar
; Univeristy of Manchester

;
; Avance II+/III Version
; Topspin 3.x

;
; \$CLASS=HighRes
; \$DIM=2D
; \$TYPE=
; \$SUBTYPE=
; \$COMMENT=

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"

"d11=30m"

"d12=20u"

"acqt0=-p1*2/PI"

"cnst30=0"

1 ze

2 d11 p1:f1

10u UNBLKGRAD

p8 ph21

;saturation pulses

p8 ph22

p18:gp8*-1

d18 BLKGRAD

50u LOCKH_OFF

1m fq=cnst29(bf ppm):f1

;solvent presaturation

d12 p19:f1

d1 cw:f1 ph29

4u do:f1

1m fq=cnst30:f1

d12 p11:f1

50u LOCKH_ON

50u UNBLKGRAMP

p1 ph1

;PROJECT

3 d20

p2 ph2

d20

p1 ph3

d20

p2 ph2
d20
lo to 3 times c
3u BLKGRAMP

go=2 ph31
d11 wr #0 if #0 ivc
50u LOCKH_OFF
lo to 1 times td1
exit

ph1=0 2
ph2=1 1 3 3
ph3=1 1 1 1 3 3 3 3
ph21=0
ph22=1
ph29=0
ph31=0 2 0 2

;POWER LEVEL
;pl0 : zero power (0W)
;pl1 : power level for pulse (default)
;pl9 : f1 channel - power level for continuous wave pulse

;PULSE DURATION
;p1 : 90 degree high power pulse
;p2 : 180 degree high power pulse
;p8 : saturation pulse [1-1.5 msec]

;GRADIENT DURATION
;p18 : saturation gradient pulse [1 msec]

;DELAY
;d1 : relaxation delay; [2-10 s]
;d11 : delay for disk I/O [30 msec]
;d18 : recovery delay for saturation gradients [1 ms]
;d20 : fixed echo time to allow elimination of J-mod. effects
; d20 should be $\ll 1/J$, but $> (50 * P2)$ [1-2 msec]

;GRADIENT SHAPE
;gpnam8 : SMSQ10.100

;GRADIENT STRENGTH
;gpz8 : saturation gradient [13%]

;CONSTANTS
;cnst29: water chemical shift (in ppm)

;OTHER
;NS: 4 * n, total number of scans: NS * TD0
;DS: 8
;td1 : number of loops defined in VCLIST (number of experiments)
;c : loop counter for T2 filter (even numbers to provide for cancellation of 180 degree pulse errors)
;define VCLIST
;vc : variable loop counter, taken from vc-list
;this pulse program produces a ser-file (PARMOD = 2D)

3. IR REST₁ with CW presaturation

```
;2drestt1irpr_cwvd
;
; T1-relaxation encoded selective TOCSY (REST)
; measurement using inversion recovery,
; using DIPSI-2 with ZQF in the TOCSY transfer
; with saturation pulses prior to d1 and
; with continuous wave excitation for water presaturation during d1 and VD;

; Modified 30/01/17
;
;      Guilherme Dal Poggetto and Laura Castañar
;      Univeristy of Manchester
;
; Avance II+/III Version
; Topspin 3.x
;
; $CLASS=HighRes
; $DIM=1D
; $TYPE=
; $SUBTYPE=
; $COMMENT=

#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>

"p2=p1*2"

"FACTOR1=(d9/(p6*115.112))/2"
"l1=FACTOR1*2"

"d11=30m"
"d12=20u"

"spoff32=0"
"spoff34=0"

"p33=1000000.0/(cnst53*4)"
"cnst33= (p33/p1) * (p33/p1)"
"spw32=plw1/cnst33"

"p35=1000000.0/(cnst54*4)"
"cnst35= (p35/p1) * (p35/p1)"
"spw34=plw1/cnst35"

"acqt0=-p1*2/3.1416"
"cnst30=0"

1 ze

2 d11 p1:f1
  10u UNBLKGRAD
  p8 ph21
  p8 ph22
  p18:gp8*-1
;saturation pulses
```

d18 BLKGRAD

50u LOCKH_OFF
1m fq=cnst29(bf ppm):f1
d12 pl9:f1
d1 cw:f1 ph29
4u do:f1
1m fq=cnst30:f1
d12 pl1:f1
50u LOCKH_ON
50u UNBLKGRAMP

;solvent presaturation

p2 ph1
4u fq=cnst29(bf ppm):f1
4u pl9:f1
vd cw:f1 ph29
4u do:f1
4u fq=cnst30:f1
4u pl1:f1

;inversion recovery with solvent saturation during vd

p1 ph2
p16:gp1
d16 pl0:f1
p12:sp12:f1 ph3
p16:gp1
d16 pl1:f1

;selective 180r

3 p1 ph4
20u gron0
(p32:sp32 ph5):f1
10u groff
100u pl10:f1

;begin DIPSI2

4 p6*3.556 ph23
p6*4.556 ph25
p6*3.222 ph23
p6*3.167 ph25
p6*0.333 ph23
p6*2.722 ph25
p6*4.167 ph23
p6*2.944 ph25
p6*4.111 ph23

p6*3.556 ph25
p6*4.556 ph23
p6*3.222 ph25
p6*3.167 ph23
p6*0.333 ph25
p6*2.722 ph23
p6*4.167 ph25
p6*2.944 ph23
p6*4.111 ph25

p6*3.556 ph25
p6*4.556 ph23
p6*3.222 ph25

p6*3.167 ph23
p6*0.333 ph25
p6*2.722 ph23
p6*4.167 ph25
p6*2.944 ph23
p6*4.111 ph25

p6*3.556 ph23
p6*4.556 ph25
p6*3.222 ph23
p6*3.167 ph25
p6*0.333 ph23
p6*2.722 ph25
p6*4.167 ph23
p6*2.944 ph25
p6*4.111 ph23
lo to 4 times l1

;end DIPSI2

p17:gp2
500u
10u gron10
(p34:sp34 ph5):f1
20u groff
10u pl1:f1
10u BLKGRAMP
p1 ph6

go=2 ph31
d11 wr #0 if #0 ivd
50u LOCKH_OFF
lo to 1 times tdl

exit

ph1= 0 0 0 0 2 2 2 2
ph2= 0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
ph3= 0 1 2 3
ph4= 0
ph6= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 1 1 1 1 1 1 1 1 1 1 1 1 1
ph5= 0
ph21=0
ph22=1
ph23=3
ph25=1
ph29=0
ph31= 0 2 0 2 0 2 0 2 2 0 2 0 2 0 2 0 1 3 1 3 1 3 1 3 3 1 3 1 3 1 3 1

;POWER LEVEL

;pl0 : zero power (0W)

;pl1 : power level for pulse (default)

;pl9 : f1 channel - power level for continuous wave pulse

;pl10 : power level for TOCSY-spinlock

;sp12 : power level of refocusing shaped pulse

;sp32 : power level of adiabatic pulse of first ZQF element

;sp34 : power level of adiabatic pulse of last ZQF element

```

;PULSE DURATION
;p1 : 90 degree high power pulse
;p2 : 180 degree high power pulse
;p6 : 90 degree low power pulse
;p8 : saturation pulse [1-1.5 msec]
;p12 : 180 degree refocusing shaped pulse
;    choose p12 according to desired selectivity
;p32 : first ZQF 180 degree inversion shaped pulse (adiabatic) [10 msec]
;p34 : second ZQF 180 degree inversion shaped pulse (adiabatic) [30 msec]

;GRADIENT DURATION
;p16 : duration of CTP gradients for selective pulse [1 msec]
;p17 : duration of homospoil z-gradients TOCSY element [1 msec]
;p18 : saturation gradient pulse [1 msec]

;DELAY
;d1 : relaxation delay; [2-10 s]
;d9 : TOCSY mixing time [50-200 ms]
;d11 : delay for disk I/O [30 msec]
;d16 : recovery delay for gradients of selective pulse CTP [400 us]
;d18 : recovery delay for saturation gradients [1 ms]

;PULSE SHAPE
;spnam12 : file name for the selective 180 refocusing shaped pulse [RSNOB or REBURP]
;spnam32 : file name for the adiabatic shaped pulse using in first ZQF [CHIRP]
;    smoothed chirp (low to high, 20% smoothing, 1000 points, 20KHz)
;spnam34 : file name for the adiabatic shaped pulse using in last ZQF [CHIRP]
;    smoothed chirp (low to high, 20% smoothing, 1000 points, 20KHz)

;GRADIENT SHAPE
;gpnam1 : SMSQ10.100
;gpnam2 : SMSQ10.100
;gpnam8 : SMSQ10.100

;GRADIENT STRENGTH
;gpz0 : first ZQF gradient [3%]
;gpz1 : CTP gradient [23%]
;gpz2 : homospoil z-gradient [31%]
;gpz8 : saturation gradient [13%]
;gpz10: last ZQF gradient [4%]

;CONSTANTS
;cnst29: water chemical shift (in ppm)
;cnst53: GammaB1 of first adiabatic ZQF shaped pulse
;cnst54: GammaB1 of last adiabatic ZQF shaped pulse

;OTHER
;NS: 16 * n, total number of scans: NS * TD0
;DS: 4
;td1 : number of delays in VDLIST (number of experiments)
;vd : variable delay, taken from vd-list
;define VDLIST
;this pulse program produces a ser-file (PARMOD = 2D)

```

4. PROJECT REST₂ with CW presaturation

```
;2drestt2projpr
;
; T2-relaxation encoded selective TOCSY (REST)
; measurement using PROJECT,
; using DIPSI-2 with ZQF in the TOCSY transfer
; with saturation pulses prior to d1 and
; with continuous wave excitation for water presaturation during d1
;
; Modified 30/01/17
```

```
;
;      Guilherme Dal Poggetto and Laura Castañar
;      Univeristy of Manchester
;
```

```
;Avance II+/III Version
;Topspin 3.x
;
```

```
;$CLASS=HighRes
;$DIM=1D
;$TYPE=
;$SUBTYPE=
;$COMMENT=
```

```
#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>
```

```
"p2=p1*2"
"d11=30m"
"d12=20u"
```

```
"FACTOR1=(d9/(p6*115.112))/2"
"l1=FACTOR1*2"
```

```
"acqt0=-p1*2/3.1416"
```

```
"spoff32=0"
"spoff34=0"
```

```
"p33=1000000.0/(cnst53*4)"
"cnst33= (p33/p1) * (p33/p1)"
"spw32=plw1/cnst33"
```

```
"p35=1000000.0/(cnst54*4)"
"cnst35= (p35/p1) * (p35/p1)"
"spw34=plw1/cnst35"
```

```
"cnst30=0"
```

```
1 ze
```

```
2 d11 p1:f1
```

```
10u UNBLKGRAD
```

```
p8 ph21
```

```
;saturation
```

```
p8 ph22
```

```
p18:gp8*-1
```

```
d18 BLKGRAD
```

50u LOCKH_OFF

1m fq=cnst29(bf ppm):f1

d12 pl9:f1

d1 cw:f1 ph29

4u do:f1

1m fq=cnst30:f1

d12 pl1:f1

50u LOCKH_ON

50u UNBLKGRAMP

p1 ph1

3 d20

;PROJECT

p2 ph2

d20

p1 ph3

d20

p2 ph2

d20

lo to 3 times c

4 p16:gp1

d16 pl0:f1

p12:sp12:f1 ph4

;selective 180r

p16:gp1

d16 pl1:f1

5 p1 ph5

20u gron0

(p32:sp32 ph7):f1

10u groff

100u pl10:f1

;begin DIPSI2

6 p6*3.556 ph23

p6*4.556 ph25

p6*3.222 ph23

p6*3.167 ph25

p6*0.333 ph23

p6*2.722 ph25

p6*4.167 ph23

p6*2.944 ph25

p6*4.111 ph23

p6*3.556 ph25

p6*4.556 ph23

p6*3.222 ph25

p6*3.167 ph23

p6*0.333 ph25

p6*2.722 ph23

p6*4.167 ph25

p6*2.944 ph23

p6*4.111 ph25

p6*3.556 ph25

p6*4.556 ph23

p6*3.222 ph25

p6*3.167 ph23
p6*0.333 ph25
p6*2.722 ph23
p6*4.167 ph25
p6*2.944 ph23
p6*4.111 ph25

p6*3.556 ph23
p6*4.556 ph25
p6*3.222 ph23
p6*3.167 ph25
p6*0.333 ph23
p6*2.722 ph25
p6*4.167 ph23
p6*2.944 ph25
p6*4.111 ph23
lo to 6 times l1

;end DIPSI2

p17:gp2
500u
10u gron10
(p34:sp34 ph7):f1
20u groff
10u p11:f1
10u BLKGRAMP
p1 ph6

go=2 ph31
d11 wr #0 if #0 ivc
50u LOCKH_OFF
lo to 1 times tdl
exit

ph1=0 0 0 0 2 2 2 2
ph2=1
ph3=1
ph4=0 1 2 3
ph5=0
ph6=0 0 0 0 0 0 0 1 1 1 1 1 1 1 2 2 2 2 2 2 2 3 3 3 3 3 3 3
ph7=0
ph21=0
ph22=1
ph23=3
ph25=1
ph29=0
ph31=0 2 0 2 2 0 2 0 1 3 1 3 3 1 3 1 2 0 2 0 0 2 0 2 3 1 3 1 1 3 1 3

;POWER LEVEL
;pl0 : zero power (0W)
;pl1 : power level for pulse (default)
;pl9 : f1 channel - power level for continuous wave pulse
;pl10 : power level for TOCSY-spinlock
;sp12 : power level of refocusing shaped pulse
;sp32 : power level of adiabatic pulse of first ZQF element
;sp34 : power level of adiabatic pulse of last ZQF element

```

;PULSE DURATION
;p1 : 90 degree high power pulse
;p2 : 180 degree high power pulse
;p6 : 90 degree low power pulse
;p8 : saturation pulse [1-1.5 msec]
;p12 : 180 degree refocusing shaped pulse
;    choose p12 according to desired selectivity
;p32 : first ZQF 180 degree inversion shaped pulse (adiabatic) [10 msec]
;p34 : second ZQF 180 degree inversion shaped pulse (adiabatic) [30 msec]

;GRADIENT DURATION
;p16 : duration of CTP gradients for selective pulse [1 msec]
;p17 : duration of homospoil z-gradients TOCSY element [1 msec]
;p18 : saturation gradient pulse [1 msec]

;DELAY
;d1 : relaxation delay; [2-10 s]
;d9 : TOCSY mixing time [50-200 ms]
;d11 : delay for disk I/O [30 msec]
;d16 : recovery delay for gradients of selective pulse CTP [400 us]
;d18 : recovery delay for saturation gradients [1 ms]
;d20 : fixed echo time to allow elimination of J-mod. effects
;    d20 should be << 1/J ,but > (50 * P2) [1-2 msec]

;PULSE SHAPE
;spnam12 : file name for the selective 180 refocusing shaped pulse [RSNOB or REBURP]
;spnam32 : file name for the adiabatic shaped pulse using in first ZQF [CHIRP]
;    smoothed chirp (low to high, 20% smoothing, 1000 points, 20KHz)
;spnam34 : file name for the adiabatic shaped pulse using in last ZQF [CHIRP]
;    smoothed chirp (low to high, 20% smoothing, 1000 points, 20KHz)

;GRADIENT SHAPE
;gpnam1 : SMSQ10.100
;gpnam2 : SMSQ10.100
;gpnam8 : SMSQ10.100

;GRADIENT STRENGTH
;gpz0 : first ZQF gradient [3%]
;gpz1 : CTP gradient [23%]
;gpz2 : homospoil z-gradient [31%]
;gpz8 : saturation gradient [13%]
;gpz10: last ZQF gradient [4%]

;CONSTANTS
;cnst29: water chemical shift (in ppm)
;cnst53: GammaB1 of first adiabatic ZQF shaped pulse
;cnst54: GammaB1 of last adiabatic ZQF shaped pulse

;OTHER
;NS: 8 * n, total number of scans: NS * TD0
;DS: 8
;td1 : number of loops defined in VCLIST (number of experiments)
;c : loop counter for T2 filter (even numbers to provide for cancellation of 180 degree pulse errors)
;define VCLIST
;vc : variable loop counter, taken from vc-list
;this pulse program produces a ser-file (PARMOD = 2D)

```

5. IR REST₁ (no presaturation)

```
;2drestt1ir
;
; T1-relaxation encoded selective TOCSY (REST)
; measurement using inversion recovery,
; using DIPSI-2 with ZQF in the TOCSY transfer
; with saturation pulses prior to d1;
;
; Modified 30/01/17
;
;      Guilherme Dal Poggetto and Laura Castañar
;      Univeristy of Manchester
;
;Avance II+/III Version
;Topspin 3.x
;
;$CLASS=HighRes
;$DIM=1D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>

"p2=p1*2"
"p3=p1"

"FACTOR1=(d9/(p6*115.112))/2"
"l1=FACTOR1*2"

"d11=30m"
"d12=20u"
"spoff32=0"
"spoff34=0"

"p33=1000000.0/(cnst53*4)"
"cnst33= (p33/p1) * (p33/p1)"
"spw32=plw1/cnst33"

"p35=1000000.0/(cnst54*4)"
"cnst35= (p35/p1) * (p35/p1)"
"spw34=plw1/cnst35"

"acqt0=-p1*2/3.1416"

"cnst30=0"

1 ze
2 d11 pl1:fl

10u UNBLKGRAD
p8 ph21
p8 ph22
p18:gp8*-1;saturation
```

d18 BLKGRAD

50u LOCKH_OFF

d1

50u LOCKH_ON

50u UNBLKGRAMP

p2 ph1

;inversion recovery

vd

p1 ph2

p16:gp1

d16 pl0:f1

p12:sp12:f1 ph3

;selective 180r

p16:gp1

d16 pl1:f1

3 p1 ph4

20u gron0

(p32:sp32 ph5):f1

10u groff

100u pl10:f1

;begin DIPSI2

4 p6*3.556 ph23

p6*4.556 ph25

p6*3.222 ph23

p6*3.167 ph25

p6*0.333 ph23

p6*2.722 ph25

p6*4.167 ph23

p6*2.944 ph25

p6*4.111 ph23

p6*3.556 ph25

p6*4.556 ph23

p6*3.222 ph25

p6*3.167 ph23

p6*0.333 ph25

p6*2.722 ph23

p6*4.167 ph25

p6*2.944 ph23

p6*4.111 ph25

p6*3.556 ph25

p6*4.556 ph23

p6*3.222 ph25

p6*3.167 ph23

p6*0.333 ph25

p6*2.722 ph23

p6*4.167 ph25

p6*2.944 ph23

p6*4.111 ph25

p6*3.556 ph23

p6*4.556 ph25


```

p6*3.222 ph23
p6*3.167 ph25
p6*0.333 ph23
p6*2.722 ph25
p6*4.167 ph23
p6*2.944 ph25
p6*4.111 ph23
lo to 4 times l1

;end DIPSI2

p17:gp2
500u
10u gron10
(p34:sp34 ph5):f1
20u groff
10u p11:f1
10u BLKGRAMP
p1 ph6

go=2 ph31
d11 wr #0 if #0 ivd
50u LOCKH_OFF

lo to 1 times td1

exit

ph1= 0 0 0 0 2 2 2 2
ph2= 0 0 0 0 0 0 0 0 2 2 2 2 2 2 2 2
ph3= 0 1 2 3
ph4= 0
ph6= 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1 1
ph5= 0
ph21=0
ph22=1
ph23=3
ph25=1
ph31= 0 2 0 2 0 2 0 2 2 0 2 0 2 0 2 0 1 3 1 3 1 3 1 3 3 1 3 1 3 1 3 1

;POWER LEVEL
;p10 : zero power (0W)
;p11 : power level for pulse (default)
;p110 : power level for TOCSY-spinlock
;sp10 : power level of double-chirp PSYCHE pulse element
;sp12 : power level of refocusing shaped pulse
;sp32 : power level of adiabatic pulse of first ZQF element
;sp34 : power level of adiabatic pulse of last ZQF element

;PULSE DURATION
;p1 : 90 degree high power pulse
;p2 : 180 degree high power pulse
;p6 : 90 degree low power pulse
;p8 : Saturation recovery pulse [1-1.5 msec]
;p12 : 180 degree refocusing shaped pulse
; choose p12 according to desired selectivity
;p32 : first ZQF 180 degree inversion shaped pulse (adiabatic) [10 msec]
;p34 : second ZQF 180 degree inversion shaped pulse (adiabatic) [30 msec]

```

;GRADIENT DURATION	
;p16 : duration of CTP gradients for Selective pulse	[1 msec]
;p17 : duration of CTP gradients for z-TOCSY	[1 msec]
;p18 : saturation recovery gradient pulse	[1 msec]
;DELAY	
;d1 : relaxation delay;	[2-10 s]
;d9 : TOCSY mixing time	[50-200 ms]
;d11 : delay for disk I/O	[30 msec]
;d16 : recovery delay for gradients of selective pulse CTP	[400 us]
;d18 : recovery delay for saturation gradients	[1 ms]
;PULSE SHAPE	
;spnam12 : file name for the selective 180 refocusing shaped pulse	[RSNOB or REBURP]
;spnam32 : file name for the adiabatic shaped pulse using in first ZQF	[CHIRP]
; smoothed chirp (low to high, 20% smoothing, 1000 points, 20KHz)	
;spnam34 : file name for the adiabatic shaped pulse using in last ZQF	[CHIRP]
; smoothed chirp (low to high, 20% smoothing, 1000 points, 20KHz)	
;GRADIENT SHAPE	
;gpnam1 : SMSQ10.100	
;gpnam2 : SMSQ10.100	
;gpnam8 : SMSQ10.100	
;GRADIENT STRENGTH	
;gpz0 : first ZQF gradient	[3%]
;gpz1 : CTP gradient	[13 or 23%]
;gpz2 : CTP gradient	[31%]
;gpz8 : saturation recovery gradient	[31%]
;gpz10: last ZQF gradient	[4%]
;CONSTANTS	
;cnst53: GammaB1 of first adiabatic ZQF shaped pulse	
;cnst54: GammaB1 of last adiabatic ZQF shaped pulse	
;OTHER	
;td1 : number of delays in VDLIST	
;vd : variable delay, taken from vd-list	
;define VDLIST	
;this pulse program produces a ser-file (PARMOD = 2D)	
;NS: 16 * n, total number of scans: NS * TD0	
;DS: 4	

6. PROJECT REST₂ (no presaturation)

```
;2drestt2proj
;
; T2-relaxation encoded selective TOCSY (REST)
; measurement using PROJECT,
; using DIPSI-2 with ZQF in the TOCSY transfer and
; with saturation pulses prior to d1;
;
; Modified 30/01/17
;
;      Guilherme Dal Poggetto and Laura Castañar
;      Univeristy of Manchester
;
;Avance II+/III Version
;Topspin 3.x
;
;$CLASS=HighRes
;$DIM=1D
;$TYPE=
;$SUBTYPE=
;$COMMENT=

#include <Avance.incl>
#include <Delay.incl>
#include <Grad.incl>

"p2=p1*2"
"d11=30m"
"d12=20u"

"FACTOR1=(d9/(p6*115.112))/2"
"l1=FACTOR1*2"

"acqt0=-p1*2/3.1416"

"spoff32=0"
"spoff34=0"

"p33=1000000.0/(cnst53*4)"
"cnst33= (p33/p1) * (p33/p1)"
"spw32=plw1/cnst33"

"p35=1000000.0/(cnst54*4)"
"cnst35= (p35/p1) * (p35/p1)"
"spw34=plw1/cnst35"

"cnst30=0"

1 ze
2 d11 pl1:fl

10u UNBLKGRAD
p8 ph21 ;saturation pulse
p8 ph22
p18:gp8*-1
d18 BLKGRAD
```

50u LOCKH_OFF
d1
50u LOCKH_ON
50u UNBLKGRAMP

p1 ph1

3 d20 ;PROJECT
p2 ph2
d20
p1 ph3
d20
p2 ph2
d20
lo to 3 times c

4 p16:gp1
d16 pl0:f1
p12:sp12:f1 ph4 ;selective 180r
p16:gp1
d16 pl1:f1

5 p1 ph5
20u gron0
(p32:sp32 ph7):f1
10u groff
100u pl10:f1

;begin DIPSI2

6 p6*3.556 ph23
p6*4.556 ph25
p6*3.222 ph23
p6*3.167 ph25
p6*0.333 ph23
p6*2.722 ph25
p6*4.167 ph23
p6*2.944 ph25
p6*4.111 ph23

p6*3.556 ph25
p6*4.556 ph23
p6*3.222 ph25
p6*3.167 ph23
p6*0.333 ph25
p6*2.722 ph23
p6*4.167 ph25
p6*2.944 ph23
p6*4.111 ph25

p6*3.556 ph25
p6*4.556 ph23
p6*3.222 ph25
p6*3.167 ph23
p6*0.333 ph25
p6*2.722 ph23

p6*4.167 ph25
p6*2.944 ph23
p6*4.111 ph25

p6*3.556 ph23
p6*4.556 ph25
p6*3.222 ph23
p6*3.167 ph25
p6*0.333 ph23
p6*2.722 ph25
p6*4.167 ph23
p6*2.944 ph25
p6*4.111 ph23
lo to 6 times l1

;end DIPSI2

p17:gp2
500u
10u gron10
(p34:sp34 ph7):f1
20u groff
10u p11:f1
10u BLKGRAMP
p1 ph6

go=2 ph31
d11 wr #0 if #0 ivc
50u LOCKH_OFF
lo to 1 times td1

exit

ph1=0 0 0 0 2 2 2 2
ph2=1
ph3=1
ph4=0 1 2 3
ph5=0
ph6=0 0 0 0 0 0 0 1 1 1 1 1 1 1 2 2 2 2 2 2 2 3 3 3 3 3 3 3
ph7=0
ph21=0
ph22=1
ph23=3
ph25=1
ph29=0
ph31=0 2 0 2 2 0 2 0 1 3 1 3 3 1 3 1 2 0 2 0 0 2 0 2 3 1 3 1 1 3 1 3

;POWER LEVEL

;pl0 : zero power (0W)
;pl1 : power level for pulse (default)
;pl10 : power level for TOCSY-spinlock
;sp12 : power level of refocusing shaped pulse
;sp32 : power level of adiabatic pulse of first ZQF element
;sp34 : power level of adiabatic pulse of last ZQF element

;PULSE DURATION

;p1 : 90 degree high power pulse
;p2 : 180 degree high power pulse

```

;p6 : 90 degree low power pulse
;p8 : saturation pulse [1-1.5 msec]
;p12 : 180 degree refocusing shaped pulse
; choose p12 according to desired selectivity
;p32 : first ZQF 180 degree inversion shaped pulse (adiabatic) [10 msec]
;p34 : second ZQF 180 degree inversion shaped pulse (adiabatic) [30 msec]

;GRADIENT DURATION
;p16 : duration of CTP gradients for selective pulse [1 msec]
;p17 : duration of homospoil z-gradients TOCSY element [1 msec]
;p18 : saturation gradient pulse [1 msec]

;DELAY
;d1 : relaxation delay; [2-10 s]
;d9 : TOCSY mixing time [50-200 ms]
;d11 : delay for disk I/O [30 msec]
;d16 : recovery delay for gradients of selective pulse CTP [400 us]
;d18 : recovery delay for saturation gradients [1 ms]
;d20 : fixed echo time to allow elimination of J-mod. effects
; d20 should be << 1/J ,but > (50 * P2) [1-2 msec]

;PULSE SHAPE
;spnam12 : file name for the selective 180 refocusing shaped pulse [RSNOB or REBURP]
;spnam32 : file name for the adiabatic shaped pulse using in first ZQF [CHIRP]
; smoothed chirp (low to high, 20% smoothing, 1000 points, 20KHz)
;spnam34 : file name for the adiabatic shaped pulse using in last ZQF [CHIRP]
; smoothed chirp (low to high, 20% smoothing, 1000 points, 20KHz)

;GRADIENT SHAPE
;gpnam1 : SMSQ10.100
;gpnam2 : SMSQ10.100
;gpnam8 : SMSQ10.100

;GRADIENT STRENGTH
;gpz0 : first ZQF gradient [3%]
;gpz1 : CTP gradient [23%]
;gpz2 : homospoil z-radiant [31%]
;gpz8 : saturation gradient [13%]
;gpz10: last ZQF gradient [4%]

;CONSTANTS
;cnst53: GammaB1 of first adiabatic ZQF shaped pulse
;cnst54: GammaB1 of last adiabatic ZQF shaped pulse

;OTHER
;NS: 8 * n, total number of scans: NS * TD0
;DS: 4
;td1 : number of loops defined in VCLIST (number of experiments)
;c : loop counter for T2 filter (even numbers to provide for cancellation of 180 degree pulse errors)
;define VCLIST
;vc : variable loop counter, taken from vc-list
;this pulse program produces a ser-file (PARMOD = 2D)

```